

Le Journal du Soft (2)

L' Oric Renifleur d' Adresses

Avec la ROM renifleuse dans votre Atmos, vous pourrez connaître les adresses de chargement des programmes protégés lorsque vous les chargerez avec CLOAD. Après un RESET, vous pourrez donc les passer directement sur disquette. Cette ROM est compatible avec Sédoric V3.0, seule version où la commande CSAVE soit déboguée. C'est le moment de ressortir vos vieilles K7 et de les passer enfin sur disquette.

Le contenu de cette ROM est proche de celui de la ROM V1.1, et pour être plus précis, de la première variante qui fut diffusée et qui présente quelques différences par rapport à la version la plus couramment répandue par la suite. Nous vous invitons à revoir notre première rubrique. Ces différences restent mineures et nous insisterons plutôt sur les modifications de code machine les plus intéressantes, celles qui permettent de connaître les adresses de chargement lors d'un CLOAD.

Que vous faut-il ?

Un peu de patience, votre moniteur favori et le fichier ATMOS.ROM (= 2^e variante) que vous avez élaboré après avoir lu notre 1^e article. En effet, bien que nous ne l'ayons pas testé, il est probable que si vous portez les modifications indiquées ci dessous dans cette deuxième variante, le résultat n'en sera que meilleur.

Sinon, si vous voulez vous trouver exactement avec la même ROM que la nôtre, utilisez la première variante. Pour cela, il vous faut ORIC.ROM, distribué avec Euphoric ou bien ATMOS.ROM que vous modifierez pour revenir à la première variante en vous basant sur notre 1^e article. Il faudra remettre en place les versions "ERROR1", "AUTO1" et "RESET1".

Pour mettre en oeuvre la ROM renifleuse, reportez-vous à ce même 1^e article. Démonstration de la ROM renifleuse lors de la visu du mois de Juin.

Modification à apporter pour obtenir une ROM renifleuse d'adresses :

1) FIXATION DES VECTEURS "READY" ET "NMI" (4 octets à changer)

Le détournement des vecteurs d'affichage du "Ready" et de la routine NMI a été couramment utilisé pour protéger les programmes. La ROM renifleuse appelle directement ces 2 routines sans passer par leur vecteur qui était situé en 001A et en 0247.

ROM normale	ROM renifleuse	commentaire
C4B4- 20 1A 00 JSR 001A	C4B4- 20 B0 CC JSR CCB0	en CCB0 se trouve la routine d'affichage
FFFA- 47 02 adr 0247	FFFA- B2 F8 adr F8B2	en F8B2 se trouve la routine de NMI

2) DETOURNEMENT VERS LA ROUTINE RENIFLEUSE (4 octets à changer)

ROM normale	ROM renifleuse	commentaire
E505- 20 6C E5 JSR E56C	E505- 20 BE FE JSR FEBE	routine 'Charger un programme' de CLOAD
E63F- 20 6C E5 JSR E56C	E63F- 20 BE FE JSR FEBE	routine 'Sauver un programme' de CSAVE

3) NOUVEAU CODE RENIFLEUR

Cet ajout affiche dans le bas à droite de l'écran la valeur de l'octet courant (en train d'être lu/écrit), l'adresse de début du fichier, l'adresse de fin du fichier, et l'adresse de l'octet courant. La valeur de l'octet courant et son adresse sont mis à jour à chaque appel de la routine et défilent donc à l'écran au fur et à mesure que progresse la lecture (ou l'écriture) du fichier. Ceci est très pratique pour faire un RESET dès que l'adresse du dernier octet est atteinte.

Le code de cette routine renifleuse est implanté en ROM dans la zone de redéfinition des caractères minuscules de "a" à "n". L'affichage de ces caractères est donc brouillé (voir "Ready", Searching..., Loading... etc), puisque ces caractères "mal redéfinis" sont recopiés en RAM à chaque boot. Si cela vous gêne, vous pouvez, soit placer ce code ailleurs (il faudra aussi modifier l'adresse d'appel indiquée au §2), soit charger en RAM un petit fichier contenant le code de définition correct des caractères altérés. Pour obtenir ce petit fichier, faites simplement un SAVE"MINUSC.CHS",A#B708,E#B7705 sur un Oric standard.

3-i) Conversion de l'adresse de l'octet courant en chaîne alphanumérique

FE80- 48	PHA	empile A
FE81- 98	TYA	
FE82- 48	PHA	empile Y
FE83- 8A	TXA	
FE84- 48	PHA	empile X
FE85- A0 20	LDY #20	index pour chaîne alphanumérique

FE87-	A5 34	LDA 34	HH de l'adresse de l'octet courant
FE89-	20 94 FE	JSR FE94	le convertit en chaîne alphanumérique
FE8C-	A5 33	LDA 33	LL de l'adresse de l'octet courant
FE8E-	20 94 FE	JSR FE94	le convertit en chaîne alphanumérique
FE91-	4C A8 FE	JMP FEA8	suite en FEA8

3-ii) Routine de conversion numérique -> alphanumérique

FE94-	48	PHA	sauve l'octet à convertir
FE95-	4A	LSR	
FE96-	4A	LSR	fait glisser le nibble de poids fort (4 bits)
FE97-	4A	LSR	à la place des 4 bits de poids faible
FE98-	4A	LSR	
FE99-	20 9D FE	JSR FE9D	traite le nibble de poids fort
FE9C-	68	PLA	puis le nibble de poids faible
FE9D-	29 0F	AND #0F	force à zéro les 4 bits de poids forts
FE9F-	AA	TAX	reste un nombre de 0 à 15, utilisé comme index
FEA0-	BD AE FE	LDA FEAE,X	pour lire un caractère dans la table
FEA3-	99 BC BF	STA BFBC,Y	sauve ce caractère à partir de BFBC + Y
FEA6-	C8	INY	incrémente Y
FEA7-	60	RTS	

3-iii) Suite du sous-programme initial

FEA8-	68	PLA	restaure X
FEA9-	AA	TAX	
FEAA-	68	PLA	restaure Y
FEAB-	A8	TAY	
FEAC-	68	PLA	restaure A
FEAD-	60	RTS	

3-iiii) Table de conversion numérique -> chaîne alphanumérique

FEAE-	30 31 32 33 34 35 36 37 38 39 41 42 43 44 45 46	correspondant à 0123456789ABCDEF
-------	---	----------------------------------

3-iiiii) Entrée principale de la routine renifleuse

33/34 contient l'adresse de l'octet courant

2A9/2AA, celle du début du fichier et

2AB/2AC celle du dernier octet du fichier.

FEBE-	20 80 FE	JSR FE80	conversion alphanumérique de l' adresse de l' octet en cours
FEC1-	A0 00	LDY #00	
FEC3-	B1 33	LDA (33),Y	lire l'octet courant
FEC5-	A0 13	LDY #13	index dans la chaîne alphanumérique
FEC7-	20 94 FE	JSR FE94	le convertit en alphanumérique
FECA-	C8	INY	incrémente l'index
FECB-	AD AA 02	LDA 02AA	entête, adresse début poids fort
FECE-	20 94 FE	JSR FE94	le convertit en alphanumérique
FED1-	AD A9 02	LDA 02A9	entête, adresse début poids faible
FED4-	20 94 FE	JSR FE94	le convertit en alphanumérique
FED7-	C8	INY	incrémente l'index
FED8-	AD AC 02	LDA 02AC	entête, adresse fin poids fort
FEDB-	20 94 FE	JSR FE94	le convertit en alphanumérique
FEDE-	AD AB 02	LDA 02AB	entête, adresse fin poids faible
FEE1-	20 94 FE	JSR FE94	le convertit en alphanumérique
FEE4-	A0 00	LDY #00	
FEE6-	4C 6C E5	JMP E56C	reprise du cours normal de CLOAD

4) MODIFICATIONS MINEURES

De ED96 à EDBF, le message de copyright :

'ORIC EXTENDED BASIC V1.1 © 1983 TANGERINE' a été transformé en :

'ORIC RENIFLEUR D'ADRESSE © 1985 SUPER S.F'

En F915 la couleur d'encre #00 (noire) a été changée en #03 (jaune).

En F91A la couleur de fond #17 (soit 23 = blanc) a été changée en #14 (soit 20 = bleu).

Et maintenant bon amusement !

André Chéramy 54 rue de Sours 28000 Chartres et Claude Sittler 11 rue de la Saône 67800 Hoenheim